

Derivazione metodi di Adams-Bashfort

```

> restart:
> with(CurveFitting);
[ArrayInterpolation, BSpline, BSplineCurve, Interactive, LeastSquares, PolynomialInterpolation,
  RationalInterpolation, Spline, ThieleInterpolation]

```

(1)

Costruzione polinomio interpolante

```

> p := 2 ;

```

$$p := 2$$

(2)

```

> XY := [seq( [x[k]-j*h, f[k-j]], j=0..p) ] ;

```

$$XY := \left[\left[x_k, f_k \right], \left[-h + x_k, f_{k-1} \right], \left[-2h + x_k, f_{k-2} \right] \right]$$

(3)

```

> Pint := PolynomialInterpolation( XY, z ) ;

```

$$\begin{aligned}
 Pint := & \frac{1}{2} \frac{(f_{k-2} - 2f_{k-1} + f_k) z^2}{h^2} \\
 & + \frac{1}{2} \frac{(3hf_k + hf_{k-2} - 4hf_{k-1} - 2f_k x_k - 2f_{k-2} x_k + 4f_{k-1} x_k) z}{h^2} \\
 & + \frac{1}{2} \frac{2h^2 f_k - 3hf_k x_k - hf_{k-2} x_k + 4hf_{k-1} x_k + f_k x_k^2 + f_{k-2} x_k^2 - 2f_{k-1} x_k^2}{h^2}
 \end{aligned}$$

(4)

```

> simplify(subs( z=x[k], Pint)) ;
simplify(subs( z=x[k]-h, Pint)) ;
simplify(subs( z=x[k]-2*h, Pint)) ;

```

$$\begin{aligned}
 & f_k \\
 & f_{k-1} \\
 & f_{k-2}
 \end{aligned}$$

(5)

```

> IntP := simplify(int( Pint, z=x[k]..x[k]+h)) ;

```

$$IntP := \frac{1}{12} h (23f_k + 5f_{k-2} - 16f_{k-1})$$

(6)

Metodo di Adams-Bashfort

```

> AB := y[k+1]=y[k]+collect(IntP, [h, f]) ;

```

$$AB := y_{k+1} = y_k + \left(\frac{23}{12} f_k + \frac{5}{12} f_{k-2} - \frac{4}{3} f_{k-1} \right) h$$

(7)